Azure DevOps,Terraform,OIDCを利用してキー管理せずにAWSにアプリケーションをデプロイする

この記事は[Azure DevOps Advent Calendar 2024](https://qiita.com/advent-calendar/2024/azuredevops)の13日目の記事です。

# はじめに
いつも会社のアドベントカレンダーには参加してますが、たまには別のアドカレにも参加してみようかなと思っていたところ、ちょうど[kkamegawa](https://x.com/kkamegawa)さんの投稿を見かけたので勢いで参加してみました。

https://x.com/kkamegawa/status/1852547515838840937

ただ、参加してみたのはいいものの私は自分のためにこんな資料を作ってしまうぐらいAzure入門者なので、最近Azure DevOps周りを触ることが増えてきたとはいえ中々ネタが決まりません。

https://speakerdeck.com/beex/awslu-xiu-zhe-notamenoazureru-men

そこで過去に書いたこのあたりの記事を元に、少し拡張した検証をしてみようと思います。

https://zenn.dev/handy/articles/azure-pipeline-shagent-ec2-ecr

で、実際に検証した構成がこちらです。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/ccd798e2-b172-6a4f-6d72-4946b2ca28d7.png)

この構成では、アプリはAzure PipelineからECRにコンテナイメージとしてPush、インフラ部分は[AWS App Runner](https://aws.amazon.com/jp/apprunner/)を利用してデプロイします。
App Runnerには自動でECRイメージの変更を検知して新しいバージョンのデプロイをしてくれる機能があるため、ECR以降のデプロイはユーザー側で操作する必要がありません。
また、アプリとインフラどちらをデプロイするかはブランチで制御を行います。(Azure PipelineとTerraform Cloudの機能を利用します)

あとこれが一番嬉しいと思うのですが、このような異なるサービス間で連携させようとした場合、デプロイ用のIAMユーザーを払い出さないといけないケースが出てくることがあり、その場合アクセスキーの管理が必要になって運用的にもセキュリティ的にもあまり推奨されません。

そこで各サービスに統合されているOpen ID Connectの仕組みを利用して、アクセスキーを利用せずにアプリ/インフラのどちらもデプロイします。
インフラのデプロイはAzure Pipelineからもやろうと思ったらできないことはないのですが、Azure Pipelineに統合されたタスクにApp Runnerへのデプロイタスクがないので、AWSCLI等でコマンドを書かないといけません。
Azure PipelineでTerraformを利用する方法でも良いのですが、それだとstateファイルの管理等を考慮する必要があり、Terraform Cloudであればstateファイルの管理を気にせずにデプロイ利用できるので採用しました。

## **余談**
私は業務内容的にクラウド周りの技術検証をよくするのですが、検証環境を毎回手で構築するのは再現性がなく手間なのでTerraformを利用することが多いです。
その際、ローカルのTerraformだと検証ごとにフォルダを分けてるので、数が多くなってくるとどれが構築済みでどれが削除済みなのか忘れてしまうことがあります。
Terraform Cloudを利用するとFree Plan(※)でも500リソースまでは無料で利用できるため、検証時のみリソースを構築して、検証が終わったら削除するようにすることでFree Planの範囲内でもかなり使えますし、Terraform Cloudコンソールからまとめて構築状況を見ることができます。
個人利用でもかなり使いやすいのでオススメです。

※[Free Organizations](https://developer.hashicorp.com/terraform/cloud-docs/overview#free-organizations)
> Free organizations are limited to 500 managed resources. Refer to What is a managed resource for more details.
> 自由組織は500の管理リソースに制限されている。 詳細については、「管理対象リソースとは」を参照してください。

# 前提
一から全てを説明するのは無理なので、以下は既に作成されている前提とします。

- Azure DevOps Organizations/Project作成済み
- Azure Reposリポジトリ作成済み
- AWSアカウント作成済み
- Terraform Cloudアカウント/Organizations/Project作成済み

# 実作業
本記事では基本的にデプロイ以外は画面操作によって構築していきます。
一部過去に類似記事を書いている手順についてはそちらを参照していただき、記事内では詳しく説明しません。
作業としては次の流れで行っていきます。

1. Azure-AWS間OIDC設定
1. Amazon ECRリポジトリ作成
1. Azure Reposリポジトリ作成
1. デプロイファイル作成
1. Azure Pipelineパイプライン作成
1. アプリデプロイ
1. Terraform Cloud Workspace作成
1. Terraform Cloud-AWS間OIDC設定
1. インフラデプロイ
1. アプリ再デプロイ
1. インフラ再デプロイ



## Azure-AWS間OIDC設定
最初にAzureとAWS間でOIDC設定を行って、AzureからAWSにデプロイするためのIAMロールを準備します。
過去に書いた以下記事で書いてますので詳しく説明まではしませんが、Azure側に若干詰まりどころがあるぐらいで、そこまで複雑な手順にはないと思います。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/7b33d04e-e858-e0d0-bb29-1651ac6d0b34.png)

https://www.beex-inc.com/blog/azure-devops-aws-deploy


## Amazon ECRリポジトリ作成
公式のドキュメントの手順に従って、コンテナイメージを格納するECRリポジトリを作成します。
リポジトリ名は「node_app」とします。

https://docs.aws.amazon.com/ja_jp/AmazonECR/latest/userguide/repository-create.html

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/4e7459fe-08d1-76b9-52b5-99494669eef6.png)

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/997df97a-4234-eb2f-8366-efd657089c98.png)


## Azure Reposリポジトリ作成
Azure Reposも同じく公式のドキュメントに従って、Reposリポジトリを作成します。
こちらもリポジトリ名を「node_app」とします。

https://learn.microsoft.com/ja-jp/azure/devops/repos/git/create-new-repo?view=azure-devops

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/0c2944fd-124c-c176-7286-ae120531e9a4.png)


## デプロイ用ファイル作成
次に作成したReposリポジトリ内にデプロイに利用するファイルを作成します。
以下のような構造になるようにファイルやフォルダを作成します。

```
.
├─ src
│  └─ default.conf
│  └─ Dockerfile
│  └─ app
│  │  └─ index.html
│  └─ infra
│     └─ provider.tf
│     └─ data.tf
│     └─ output.tf
│     └─ apprunner.tf
└─ README.md
```

README.md以外のファイルの中身を以下に記載します。

```config:default.conf
server {
        listen 80 default_server;
        listen [::]:80 default_server;

        root /app;

        location / {
        }
}
```

```Dockerfile:Dockerfile
FROM alpine:3.6

# nginxのインストール
RUN apk update && \
    apk add --no-cache nginx

# ドキュメントルート
ADD app /app
ADD default.conf /etc/nginx/conf.d/default.conf

# ポート設定
EXPOSE 80

RUN mkdir -p /run/nginx

# フォアグラウンドでnginx実行
CMD nginx -g "daemon off;"
```

```html:app/index.html
<html>
<body>
TEST
</body>
</html>
```

```terraform:infra/provider.tf
terraform {
  required_providers {
    aws = {
      source  = "hashicorp/aws"
      version = "~> 4.0"
    }
  }
}

provider "aws" {
  region = "ap-northeast-1"
}
```

```terraform:infra/data.tf
data "aws_ecr_repository" "node_app" {
  name = "node_app"
}
```

```terraform:infra/output.tf
output "service_url" {
  value = aws_apprunner_service.example.service_url
}

output "service_status" {
  value = aws_apprunner_service.example.status
}
```

```terraform:infra/apprunner.tf
# IAM Role for App Runner
resource "aws_iam_role" "apprunner_service_role" {
  name = "apprunner-service-role"

  assume_role_policy = jsonencode({
    Version = "2012-10-17"
    Statement = [
      {
        Action = "sts:AssumeRole"
        Effect = "Allow"
        Principal = {
          Service = "build.apprunner.amazonaws.com"
        }
      }
    ]
  })
}

# IAM Role Policy Attachment
resource "aws_iam_role_policy_attachment" "apprunner_service_role_policy" {
  role       = aws_iam_role.apprunner_service_role.name
  policy_arn = "arn:aws:iam::aws:policy/service-role/AWSAppRunnerServicePolicyForECRAccess"
}

# App Runner Service
resource "aws_apprunner_service" "node_app" {
  service_name = "nginx-service"

  source_configuration {
    image_repository {
      image_configuration {
        port = "80"  # Nginxのポート
        runtime_environment_variables = {
          "NGINX_WORKER_PROCESSES" = "auto"
        }
      }
      image_identifier      = "${data.aws_ecr_repository.node_app.repository_url}:latest"
      image_repository_type = "ECR"
    }
    authentication_configuration {
      access_role_arn = aws_iam_role.apprunner_service_role.arn
    }
    auto_deployments_enabled = true
  }

  instance_configuration {
    cpu    = "1024"
    memory = "2048"
  }

  health_check_configuration {
    path = "/"
    protocol = "HTTP"
    healthy_threshold = 1
    interval = 10
    timeout = 5
    unhealthy_threshold = 5
  }

  tags = {
    Environment = "dev"
    Name        = "nginx-service"
  }
}
```

## Azure Pipelineパイプライン作成
Azure DevOpsコンソールからAzure Pipeline画面を開き、「New pipeline」から新しいパイプラインを作成します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/2496ed7b-8afa-a096-c14a-d5bca3265ce0.png)

Azure Repos Gitを選択します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/bd14b816-7e82-0c05-46a6-0d2629493164.png)

先ほど作成した「node_app」を選択します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/5388288d-f58f-307a-1122-f649672c2331.png)

後で上書きするので正直どれでも良いですが「Docker」を選択します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/f3911dd3-02d7-40e0-ac31-0c2dbf81b62a.png)

Dockerfileが格納されている場所が表示されているので、Reposリポジトリ内のDockerfileパスと一致しているか確認します。
「$(Build.SourcesDirectory)」はBuild時のカレントディレクトリで、Reposリポジトリのルートディレクトリと思ってもらってOKです。
問題なければ「Validate and configure」を押下します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/7d54a4f1-0cd1-4c85-ae62-9d310391409f.png)

Azure Pipelineが参照するpipeline YAMLが自動で作成されます。
先ほどDockerを選択したので、Dcokerfileの内容を元にBuild Imageするタスクがデフォルトで作成されています。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/9f1d7f8e-895b-ff58-b9b0-0c9ee2de1aeb.png)

上記pipeline YAMLの内容は全選択→削除し、以下の内容を貼り付けてパイプラインを作成します。

```yml:azure-pipeline.yml
# Docker
# Build a Docker image
# https://docs.microsoft.com/azure/devops/pipelines/languages/docker

trigger:
- release

pool:
  vmImage: ubuntu-latest

resources:
- repo: self

variables:
  tag: '$(Build.BuildId)'

stages:
- stage: Build
  displayName: Build image
  jobs:
  - job: Build
    displayName: Build
    steps:
    - task: Docker@2
      displayName: Build an image
      inputs:
        command: build
        dockerfile: '$(Build.SourcesDirectory)/src/Dockerfile'
        buildContext: '$(Build.SourcesDirectory)/src'
        repository: $(DOCKER_REPOSITORY_NAME)

    - task: ECRPushImage@1
      inputs:
        awsCredentials: 'aws-oidc-user2'
        regionName: 'ap-northeast-1'
        imageSource: 'imagename'
        sourceImageName: $(DOCKER_REPOSITORY_NAME)
        sourceImageTag: $(Build.BuildId)
        pushTag: latest
        repositoryName: $(DOCKER_REPOSITORY_NAME)
```

2つだけ特筆事項があります。
まず1つ目が以下の変数設定です。

> $(DOCKER_REPOSITORY_NAME)

この変数は作成したパイプラインに紐付く変数で、パイプラインの編集画面の「Variables」から設定できます。
今回は次のように設定します。設定内容は最初に作成したECRリポジトリの名前にします。(どちらも"node_app"にしたのでわかりづらいとは思いますが。。)

- Name：DOCKER_REPOSITORY_NAME
- Value：node_app

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/037d7cad-7711-cd42-de09-208dbdd90cc8.png)

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/08abe922-b947-b293-5a4f-d5e5c6d4ce7a.png)

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/cc57e110-6bdb-43c5-4c53-81642ad96f3f.png)

2つ目が「task: ECRPushImage@1」タスクの以下の部分です。

> awsCredentials: 'aws-oidc-user2'

ここではAzure-AWS間OIDC設定で作成したAzure DevOpsのService connections内のコネクション名を指定する必要があります。

https://www.beex-inc.com/blog/azure-devops-aws-deploy#heading9

Service connectionsはProject Settingsから確認ができます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/09e70e5f-488a-8d51-18bd-7cd32c06f27c.png)

うまく設定でできていれば次のような表示になっているはずです。
Role to AssumeはAzureからAWSにデプロイするのに利用するIAMロールARNなので、ご自身の環境で構築する際はよしなに読み替えてください。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/ee7dcdcf-7d93-9968-d4db-50df06001d88.png)

ここまででパイプラインの作成は完了です。

## アプリデプロイ
まだインフラを構築していないのでアプリケーションは実行されませんが、一旦コンテナイメージが正常にPushされるか確認してみます。
Azure Pipelines画面から作成した「node_app」パイプラインを開いて、「Run pipeline」を実行します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/3ac60ecb-51b4-ba9c-b273-38ad5a4c2dd5.png)

設定や実装が上手くいっていればECRPushImageタスクが完了するはずです。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/97b08892-eb42-28d2-a0be-7917eb870aec.png)

AWSマネジメントコンソールから確認するとコンテナイメージが1つ増えていることがわかります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/0cd41dc4-c6cd-efaa-a0ba-317bacd0e72a.png)

Azure Pipelineパイプラインを作成する際に、以下の設定を入れてパイプラインの実行トリガーをreleaseブランチに変更が入った場合に実行されるように変更したので、Azure Reposからreleaseブランチを作成しておきます

```yml
trigger:
- release
```

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/6f132352-8f05-1a29-b0ec-f8857149f784.png)

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/13546833-42f9-620a-49a9-3c90b5b9665c.png)

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/32c55fc8-537d-f000-7e02-2aeae420c657.png)

インフラ構築時に利用するため、release_infraブランチも同じ要領で作っておきます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/1397eac0-3abc-1457-b886-1e9212628219.png)


## Terraform Cloud Workspace作成
アプリ側の実装・動確が終わったので、次はインフラデプロイのための環境を構築していきます。
[Terraform Cloud](https://app.terraform.io/)にログインして以下の流れでWorkspaceを作成します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/afe1d5ae-3c9a-30c2-cb8c-9b2cc90dfe91.png)

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/b6c31f4c-ef55-34c4-7253-53cfef36d0ca.png)

今回はAzure DevOpsと連携するので「Version Control Workflow」を選択します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/7cf940b1-33b0-e0df-3009-208704b1a58d.png)

接続するProviderを聞かれるので、「Azure DevOps Service」を選択します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/a0567b68-c9ea-6a14-1f77-10c8e98be1a8.png)

以下のドキュメントを参考にしつつTerraform CloudとAzure DevOps Servicesの接続設定をします。

- [Configuring Azure DevOps Services Access](https://developer.hashicorp.com/terraform/cloud-docs/vcs/azure-devops-services)

必要なのは次の5ステップになります。

1. Azure DevOps Services Profileから、新しいアプリケーションを作成する
1. HCP Terraformで、プロバイダを設定する
1. HCP Terraformで、高度な設定を構成する (オプション)
1. 対象リポジトリを指定する
1. Workspaceの詳細設定をする

### 事前設定
Terraform Cloudとの接続設定をする前に、Azure DevOps側で次の作業を行う必要があります。

Organization Settings>Policies>Application connection policiesから「Third-party application access via OAuth」を有効化します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/5a060ae1-3b06-1084-b9a4-842389cff3f8.png)

### Azure DevOps Services Profileから、新しいアプリケーションを作成する
まずはAzure DevOps側でTerraform Cloudのアプリケーションを登録します。
作成画面にURLが2つあるので、どちらかをコピーしてAzure DevOpsにログイン済みのブラウザの別タブで開きます。
※私の環境では上のほうのみアクセスできました

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/b053f3ab-f8ca-4fad-e9ae-c2942b5ae251.png)

アクセスすると次のような画面が表示されます。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/5bcac85d-8a6b-fa06-c8af-a974bec4344d.png)

先ほどのTerraform Cloud側の画面に必要な情報が載っているため、この情報を転記していきます。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/89ea5956-951e-3044-1be1-a0ae5f44496a.png)

一旦必須のところだけ入力してページ下部にある「Create Application」を押下します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/6f835992-f9ca-ec56-f0e0-de4b074cd743.png)

すると次のようなページに遷移するので、次の手順でApp IDとClient Secretを利用します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/733e6e69-f6ea-feef-ad63-549994e41246.png)

### HCP Terraformで、プロバイダを設定する
Terraform Cloudコンソールに戻って、先ほど確認したApp IDとClient Secretを転記します。
Nameはオプションなので必要であればわかりやすい名前を付けます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/3fbe31c1-3f16-637c-f5db-69d35be8a207.png)

たまに半角スペースが入り込んでエラーが出たりするのでご注意ください。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/6f27eeba-f59c-7079-640e-4ed99807b1bc.png)

「Connect and continue」を押下するとTerraform CloudがAzure DevOpsに対してアクセスリクエストを送るので、Azure DevOps側でそのリクエストを承認します。
Terraform CloudとAzure DevOpsの設定を別ブラウザで行っていた場合は、次のようにMSアカウントのサインイン画面が表示されることがあります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/27c2a2e3-1195-9730-b891-327dae7366af.png)

その場合はURLを丸っとコピーして対象のブラウザに貼り付けすると以下の画面が表示されると思います。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/64e147bc-57f0-bdc5-dfa4-e8fee58ac895.png)

上記画面でAceeptを押下してリクエストを許可します。

### HCP Terraformで、高度な設定を構成する (オプション)
オプションとして、Azure DevOpsからアクセスできる対象のWorkspaceの指定やPolicyやModules等の有効化やSSHキーペアの設定ができます。
今回はプロジェクトの指定だけをして「Skip and finish」を押下します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/0a428188-6097-f988-8cab-fd1cb88a331d.png)

### 対象リポジトリを指定する
Workspaceを連携するリポジトリを選択します。
今回は前の手順で作成した「node_app」リポジトリを選択します。
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/9ab3918a-67ad-9c24-fc19-2fe20898b111.png)

### Workspaceの詳細設定をする
最後にWorkspaceの詳細設定を行います。

Workspace Nameにはわかりやすく「nodeapp_ws」という名前を入力します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/256d69c7-07e6-94c5-b79b-1197a860b047.png)

この画面ではWorkspaceを実行するトリガーやオートApply等の設定が可能です。

以下ではTerraformを実行するWorking Directoryを指定することができるので、Azure Reposリポジトリの構成からTerraformコードが格納されているパス「src/infra」を記載します。
Auto-applyを設定すると自動でデプロイされてしまうため、今回は手動でApply操作をするためにチェックはなしにします。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/7a93f07e-a2fc-f0e7-4162-63f80931cd12.png)

次はTerraformをどのようにトリガーするかを設定します。
ここではBranchベーストリガーを選択し、アプリとブランチを分けるために「release_infra」を記載します。
また、トリガーを特定のファイルが更新されたタイミングで実行されるように設定も可能ですが、ここでは指定したリポジトリ/フォルダ配下が更新されたらトリガーされるようにします。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/cf6a7b87-cdb9-61c7-bb6f-b08242d36544.png)

最後はサブモジュールをクローリングするかの設定なので、チェックなしのまま「Create」を押下してWorkspaceを作成します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/f9e6de61-cf64-bf7b-17c0-b464829bd134.png)

以下が表示されたらWorkspaceの作成は完了です。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/ff0de729-18a4-fa17-ff79-7fc2f4a1db06.png)

## Terraform Cloud-AWS間OIDC設定
WorkspaceができたらTerraform CloudとAWS間のOIDC設定を行って、AWSへのデプロイができるようにします。
AWSとのOIDC設定手順は以下ドキュメントに記載があります。

https://developer.hashicorp.com/terraform/cloud-docs/workspaces/dynamic-provider-credentials/aws-configuration

次の3ステップで設定していきます。

1. OIDCアイデンティティプロバイダの作成
1. IAMロールの作成
1. Terraform Cloud Workspaceに変数を設定

### OIDCアイデンティティプロバイダの作成
基本的にはドキュメントの以下の箇所を確認すればOIDC ID Providerを作成できると思います。

https://docs.aws.amazon.com/IAM/latest/UserGuide/id_roles_providers_create_oidc.html#manage-oidc-provider-console

次の設定を行います。設定値は[こちら](https://developer.hashicorp.com/terraform/cloud-docs/workspaces/dynamic-provider-credentials/aws-configuration#configure-aws)に記載されています。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/2cda3ef6-140b-02d1-ba08-b8aa5181e69b.png)

作成できると以下のようになります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/922206cb-d294-a3df-9db2-42220785f354.png)

### IAMロールの作成
次に作成したID Providerに紐付くIAMロールを作成します。
デプロイ時に利用されるIAMロールになりますので、権限を制御したい場合はこのロールのポリシーを制限する必要があります。

ID Providerページの「Assign role」を押下して、「Create a new role」を選択します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/1c402445-1b29-dcf9-edbf-2354c42dfb14.png)

IAMロールの信頼ポリシーを設定します。
次の内容で設定します。

- Identity provider：app.terraform.io
- Audience：aws.workload.identity
- Organization：[Terraform Cloudの該当Organization名]
- Project：[Terraform Cloudの該当のProjecto名]　※今回は * で記載
- Workspace：[新規作成したTerraform Cloud Workspace名]
- Run Phase：[Plan or Apply]　※今回は * で記載

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/607c1f93-e0bd-fa6c-27d1-f1ac83ad4c9a.png)

今回は検証なのでAdministratorAccessを付与します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/5ea6be94-1d34-f6ea-3f1e-6d784acef1d6.png)

IAMロールの名前を入力して「create role」を押下します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/c7ccdfb8-66b1-29bf-7128-512a7082817b.png)

作成できたら、IAMロールのARNの値をメモしておきます。(後で使います)

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/8669f161-1ea3-cd25-811d-dc03468540ea.png)

### Terraform Cloud Workspaceに変数を設定
作成したWorkspaceのVariablesページを開いて「Workspace variables」を以下のように設定します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/3c413353-3d76-92eb-032d-4295413dab1c.png)

変数設定時に「Terraform variable」と「Environment variable」の2つのカテゴリがありますが、「Environment variable」を利用します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/419b4045-7b1f-99a0-2519-aca68020e422.png)

ここまでできたらTerraform CloudからAWSにデプロイする設定は完了です。

## インフラデプロイ
試しに手動で動かすため、Azure Reposのrelease_infraブランチを開いて、src/infra/provider.tfを以下のように修正してCommitします。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/84b44ea2-cff6-bb6e-d15f-c88fdb53f3ee.png)

```terraform:src/infra/provider.tf
terraform {
  required_version = ">= 0.14"
  required_providers {
    aws = {
      source  = "hashicorp/aws"
      version = "~> 5.0"
    }
  }
}

provider "aws" {
  region = "ap-northeast-1"
}
```

うまく動かなかったので、WorkspaceのVersion Controlの「Automatic speculative plans」のチェックを外して保存します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/4e56a0bd-e72e-a7af-0bfb-6d10b6514076.png)

output.tfを空にしてCommitしてみます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/c09278ca-b5b0-6b36-2b8d-63490cca2b3e.png)

WorkspaceのRunsページを見ると処理が開始され、Planまで実行されていることがわかります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/8ecf1f90-e3f9-cd89-e61b-63b52c318ad1.png)

「Confirm & apply」を押下します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/d7048655-ee5e-4895-6329-c71a45f8b608.png)

Applyが始まってデプロイが開始しました。
App Runnerは若干構築に時間がかかります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/ffdd789f-bc2e-fd00-0699-5471ee9b0b89.png)

Applyが成功しました。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/397b8855-1dcd-66c8-3da1-0660873f369b.png)

AWSマネジメントコンソールからApp Runnerコンソールを見るとリソースが作られていることが確認できました。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/1f38b38b-037a-64a2-0ffb-4c60233e6399.png)

App RunnerはアプリケーションにアクセスできるURLを「Default domain」で払い出されるので、ブラウザからアクセスしてみるとちゃんとindex.htmlに実装したTESTという文字が表示されていることがわかります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/93dbaad0-fe1a-6da5-7d9d-6f51ba77d6b9.png)

インフラのデプロイも成功しましたので、これでアプリとインフラのデプロイパイプラインの構築が完了しました。

## アプリ再デプロイ
デプロイパイプラインが作れたので、コードを変更して自動で反映されるか確認してみます。
また、アプリとインフラの更新がそれぞれのパイプラインに影響しないことも確認します。

まずはアプリ側のファイルの更新を行います。
releaseブランチに移動後、index.htmlの"TEST"を"TEST Update"に修正してCommitします。
※本来であればmainブランチからPR・マージするべきですが今回は手順を省略します。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/3398b981-7220-e880-3422-f4033f1c158c.png)

対象のAzure Pipelineのパイプラインが実行されていることが確認できます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/245c33ce-bd27-4307-d478-efac304438c9.png)

ECRリポジトリを見るとイメージが更新されていました。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/40348dc0-b71c-4d3a-3b26-4abd6fcd1bac.png)

App Runnerは対象のECRリポジトリをウォッチしているので、更新があれば自動でデプロイが走っていることがわかります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/09d954a8-f886-e4ae-4bb3-9e3d22644d75.png)

修正した内容がちゃんと反映されました。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/157d21ee-b4c8-863f-fcc9-99b02d19fbc3.png)

Terraform Cloudのほうは実行されていないことも確認できます。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/24cdc706-7697-43ad-8bbf-8602d4037fb0.png)

無事アプリについてはデプロイパイプラインが動きました。

## インフラ再デプロイ
今度はインフラのデプロイパイプラインを動かします。
構築したApp RunnerにはWAFが紐付けられていないので、WAFを作成してApp Runnerと紐付けるようにします。
また、WAFで自身のIPアドレス以外からのアクセスを拒否します。
release_infraブランチに移動して以下の修正を入れてCommitします。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/65f51179-e238-6add-207a-466418566b2e.png)

```terraform:apprunner.tf
# IAM Role for App Runner
resource "aws_iam_role" "apprunner_service_role" {
  name = "apprunner-service-role"

  assume_role_policy = jsonencode({
    Version = "2012-10-17"
    Statement = [
      {
        Action = "sts:AssumeRole"
        Effect = "Allow"
        Principal = {
          Service = "build.apprunner.amazonaws.com"
        }
      }
    ]
  })
}

# IAM Role Policy Attachment
resource "aws_iam_role_policy_attachment" "apprunner_service_role_policy" {
  role       = aws_iam_role.apprunner_service_role.name
  policy_arn = "arn:aws:iam::aws:policy/service-role/AWSAppRunnerServicePolicyForECRAccess"
}

# App Runner Service
resource "aws_apprunner_service" "node_app" {
  service_name = "nginx-service"

  source_configuration {
    image_repository {
      image_configuration {
        port = "80"  # Nginxのポート
        runtime_environment_variables = {
          "NGINX_WORKER_PROCESSES" = "auto"
        }
      }
      image_identifier      = "${data.aws_ecr_repository.node_app.repository_url}:latest"
      image_repository_type = "ECR"
    }
    authentication_configuration {
      access_role_arn = aws_iam_role.apprunner_service_role.arn
    }
    auto_deployments_enabled = true
  }

  instance_configuration {
    cpu    = "1024"
    memory = "2048"
  }

  health_check_configuration {
    path = "/"
    protocol = "HTTP"
    healthy_threshold = 1
    interval = 10
    timeout = 5
    unhealthy_threshold = 5
  }

  tags = {
    Environment = "dev"
    Name        = "nginx-service"
  }
}

# WAF Web ACL
resource "aws_wafv2_web_acl" "apprunner_waf" {
  name        = "apprunner-waf"
  description = "WAF for App Runner service with IP whitelist"
  scope       = "REGIONAL"

  default_action {
    block {}  # デフォルトですべてのトラフィックをブロック
  }

  # IP ホワイトリストルール
  rule {
    name     = "AllowSpecificIP"
    priority = 0  # 最優先で評価

    action {
      allow {}
    }

    statement {
      ip_set_reference_statement {
        ip_set_forwarded_ip_config {
          fallback_behavior = "MATCH"
          header_name       = "X-Forwarded-For"
          position         = "FIRST"
        }
        arn = aws_wafv2_ip_set.whitelist.arn
      }
    }

    visibility_config {
      cloudwatch_metrics_enabled = true
      metric_name               = "AllowSpecificIPMetric"
      sampled_requests_enabled  = true
    }
  }

  visibility_config {
    cloudwatch_metrics_enabled = true
    metric_name               = "AppRunnerWafMetric"
    sampled_requests_enabled  = true
  }
}

# IP セットの定義
resource "aws_wafv2_ip_set" "whitelist" {
  name               = "apprunner-whitelist"
  description        = "Whitelisted IP addresses"
  scope              = "REGIONAL"
  ip_address_version = "IPV4"
  addresses          = ["[自身のIPアドレス]/32"]
}

# WAF Web ACL Association
resource "aws_wafv2_web_acl_association" "apprunner_waf_association" {
  resource_arn = aws_apprunner_service.node_app.arn
  web_acl_arn  = aws_wafv2_web_acl.apprunner_waf.arn
}
```

Planは成功したので次はApplyします。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/cf302260-3103-3efc-6070-5d713c276c50.png)

1リソースのみエラーが出ました。
WAFのログの設定で失敗しているようです。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/bfd39d87-0b97-659f-fa53-befe809cf3b6.png)

今回は必須ではないので、apprunner.tfファイル内の以下の箇所を削除して再度Commitします。

```
# Logging Configuration for WAF
resource "aws_cloudwatch_log_group" "waf_log_group" {
  name              = "/aws/waf/apprunner"
  retention_in_days = 30
}

resource "aws_wafv2_web_acl_logging_configuration" "waf_logging" {
  log_destination_configs = [aws_cloudwatch_log_group.waf_log_group.arn]
  resource_arn           = aws_wafv2_web_acl.apprunner_waf.arn
}
```

Applyが完了しました。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/df6baa61-9118-fc23-29e3-470d3bab3962.png)

App Runnerリソースを確認すると先ほど作成したWAFが設定されていることがわかります。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/609b6eaf-538b-8671-5438-3ca80815d256.png)

試しにIPアドレスを変えてアクセスしてみると想定通りアクセスが拒否されました。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/fd0c7a1d-42dc-220e-17f5-d9bcb582a9fa.png)

ECRリポジトリのコンテナイメージも増えていないので、アプリ側のパイプラインも動いてなさそうです。

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/440403/9c5d82e5-4228-7ec8-bb8e-1ba4c0febadb.png)

これでインフラについても正常にデプロイパイプラインが動きました。

# おわりに
担当日という締め切りに追われて思いつきでネタを決めましたが、思ったよりもボリュームが多くて若干後悔しました。
Azure DevOps、Terraform、OIDC、AWSという闇鍋みたいな組み合わせを使う人がどれだけいるんだろうかと思いつつ、部分部分の設定手順は参考にはなるんじゃないかとは思ってます。

この記事がどなたかの参考になれば幸いです。